Examples of factoring radical ideals into prime components.

```
> restart;
```

```
> interface(imaginaryunit=_i):
  with(PolynomialIdeals):
```

This is the example from section 3.1 on page 112.

```
> I := <x^2+y+z-1,x+y^2+z-1,x+y+z^2-1>;
```

$$I := \langle z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle$$

```
> IsRadical(I), IsPrime(I), IsZeroDimensional(I);
```

$$false,\ false,\ true$$

```
> G := Groebner[Basis](I,plex(x,y,z));
```

$$G := [z^6 - 4\,z^4 + 4\,z^3 - z^2,\ z^4 + 2\,y\,z^2 - z^2,\ y^2 - z^2 - y + z,\ z^2 + x + y - 1]$$

```
> f := factor(G[1]);
```

$$f := z^2\,(z^2 + 2\,z - 1)\,(-1 + z)^2$$

One way to try to compute the radical is to throw f_{red} the square-free part of this polynomial into I

```
> I := <I,z*(z^2+2*z-1)*(z-1)>;
```

$$I := \langle z\,(z^2 + 2\,z - 1)\,(-1 + z),\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle$$

```
> IsRadical(I), IsPrime(I), IsZeroDimensional(I);
```

$$true,\ false,\ true$$

```
> G := Groebner[Basis](I,plex(x,y,z));
```

$$G := [z^4 + z^3 - 3\,z^2 + z,\ z^3 + 2\,y\,z - z,\ y^2 - z^2 - y + z,\ z^2 + x + y - 1]$$

```
> f := factor(G[1]);
```

$$f := z\,(z^2 + 2\,z - 1)\,(-1 + z)$$

Now we are going to split I into three components P_1, P_2, P_3 corresponding to the three factors of f and then we will have $I = P_1 \cap P_2 \cap P_3$.

```
> P1 := Quotient(I,<quo(f,z,z)>);
  P2 := Quotient(I,<quo(f,z-1,z)>);
  P3 := Quotient(I,<quo(f,z^2+2*z-1,z)>);
```

$$P1 := \langle z,\ -1 + x + y,\ y^2 - y\rangle$$

$$P2 := \langle x,\ y,\ -1 + z\rangle$$

$$P3 := \langle y - z,\ -z + x,\ z^2 + 2\,z - 1\rangle$$

We could have done this by putting the three irreducible factors of f into the basis for I this way

```
> Groebner[Basis](<I,z>,plex(x,y,z));
  Groebner[Basis](<I,z-1>,plex(x,y,z));
  Groebner[Basis](<I,z^2+2*z-1>,plex(x,y,z));
```

$$[z,\ y^2 - y,\ -1 + x + y]$$

$$[-1 + z,\ y,\ x]$$

$$[z^2 + 2\,z - 1,\ y - z,\ -z + x]$$

```
> IsPrime(P1), IsPrime(P2), IsPrime(P3);
```

$$\mathit{false}, \mathit{true}, \mathit{true}$$

```
> G := factor(Groebner[Basis](P1,plex(x,y,z)));
```

$$G := [z, y\,(-1+y), -1+x+y]$$

```
> P4, P5 := Quotient(P1,<y>), Quotient(P1,<y-1>);
```

$$P4, P5 := \langle x, z, -1+y\rangle, \langle y, z, x-1\rangle$$

```
> IsPrime(P4), IsPrime(P5);
```

$$\mathit{true}, \mathit{true}$$

We are done: the prime decomposition of $\sqrt{I}$ is the following four components

```
> P2, P3, P4, P5;
```

$$\langle x, y, -1+z\rangle, \langle y-z, -z+x, z^2+2\,z-1\rangle, \langle x, z, -1+y\rangle, \langle y, z, x-1\rangle$$

```
> J := Intersect(P2,P3,P4,P5);
```

$$J := \langle -x\,z+y\,z, x\,y-x\,z, z^2+x+y-1, y^2+x+z-1, x^2+y+z-1\rangle$$

```
> Groebner[Basis](J,plex(x,y,z));
```

$$[z^4+z^3-3\,z^2+z, z^3+2\,y\,z-z, y^2-z^2-y+z, z^2+x+y-1]$$

The above example worked because when we computed the Groebner basis for I we found a polynomial that factored. This does not always happen. Consider the following example

```
> I := <x^2+1,y^2+1,z^2+1>;
```

$$I := \langle x^2+1, y^2+1, z^2+1\rangle$$

```
> G := Groebner[Basis](I,plex(x,y,z));
```

$$G := [z^2+1, y^2+1, x^2+1]$$

Notice that G is a Groebner basis for I in every monomial ordering since LT(x^2+1) = x^2 in every monomial ordering. But it is not prime over the field of rational numbers Q.

```
> IsPrime(I), IsRadical(I);
```

$$\mathit{false}, \mathit{true}$$

We can solve this problem by inspection. The second generator minus the first factor.

```
> f := (x^2+1)-(y^2+1);
```

$$f := x^2-y^2$$

```
> factor(f);
```

$$(x-y)\,(x+y)$$

```
> P1, P2 := Quotient(I,<x-y>), Quotient(I,<y+x>);
```

$$P1, P2 := \langle x+y, x^2+1, z^2+1\rangle, \langle -x+y, x^2+1, z^2+1\rangle$$

Again, the second generator minus the third factors into $(y-z)\,(y+z)$ so we repeat this decomposition on each component.

```
> P11, P12, P21, P22 := Quotient(P1,<y-z>), Quotient(P1,<y+z>),
                        Quotient(P2,<y-z>), Quotient(P2,<y+z>);
```

$$P11, P12, P21, P22 := \langle x+y, z-x, x^2+1\rangle, \langle x+y, x+z, x^2+1\rangle, \langle -x+y, x+z, x^2+1\rangle, \langle -x+y, z-x, x^2+1\rangle$$

```
> map( IsPrime, [P11, P12, P21, P22] );
```

$$[true, true, true, true]$$

```
> Intersect(P11,P12,P21,P22);
```

$$\langle x^2+1,\ y^2+1,\ z^2+1\rangle$$

But how would we do this if we cannot "spot" a polynomial in I that factors ?

One approach is to make a linear subsitution to try to "put the ideal in general position" so that when we compute the lex Groebner basis with $x > y > z$ we get the form [$f(z)$, $x - g(z)$, $y - h(z)$] from where the problem is easily solved - we just need to look at the factors of $f(z)$ since the other polynomials are LINEAR in x and y. This idea only works if the ideal is zero dimensional, i.e. the variety of the ideal has finitely many solutions.

```
> J := < subs( z=z+5*x-3*y, Generators(I) ) >;
```

$$J := \langle (z+5\,x-3\,y)^2+1,\ x^2+1,\ y^2+1\rangle$$

```
> G := Groebner[Basis](J,plex(x,y,z));
```

$$G := [z^8+140\,z^6+5278\,z^4+40860\,z^2+35721,\ 19\,z^7+2849\,z^5+123529\,z^3+1161216\,y+1281915\,z,\ 115\,z^7+16289\,z^5+636265\,z^3+5806080\,x+6426171\,z]$$

```
> f := factor(G[1]);
```

$$f := (z^2+9)\,(z^2+49)\,(z^2+81)\,(z^2+1)$$

```
> S1 := Simplify(<op(1,f),J>);
```

$$S1 := \langle -x+y,\ z+3\,x,\ x^2+1\rangle$$

Undoing the linear change of variables we get the first prime component of I

```
> P1 := Simplify( <subs(z=z-5*x+3*y,Generators(S1))> );
```

$$P1 := \langle -x+y,\ x+z,\ x^2+1\rangle$$

```
> for k from 2 to 4 do
      P||k := Simplify( <subs(z=z-5*x+3*y,Generators(<op(k,f),J>))> );
  od;
```

$$P2 := \langle x+y,\ z-x,\ x^2+1\rangle$$

$$P3 := \langle x+y,\ x+z,\ x^2+1\rangle$$

$$P4 := \langle -x+y,\ z-x,\ x^2+1\rangle$$

These components are all prime over the field of rational numbers. A check

```
> Simplify( Intersect(P1,P2,P3,P4) );
```

$$\langle x^2+1,\ y^2+1,\ z^2+1\rangle$$

Simplify computes a Groebner basis for an ideal in the graded ordering by default. The PrimeDecomposition command computes the prime decomposition of the radical of I.

```
> Primes := [PrimeDecomposition(I)];
```

$$Primes := [\langle x+z,\ y-z,\ x^2+1,\ y^2+1,\ z^2+1\rangle,\ \langle x+z,\ y+z,\ x^2+1,\ y^2+1,\ z^2+1\rangle,\ \langle y-z,\ -z+x,\ x^2+1,\ y^2+1,\ z^2+1\rangle,\ \langle y+z,\ -z+x,\ x^2+1,\ y^2+1,\ z^2+1\rangle]$$

```
> map(Simplify,Primes);
```

$$[\langle x+z,\ y-z,\ z^2+1\rangle,\ \langle x+z,\ y+z,\ z^2+1\rangle,\ \langle y-z,\ -z+x,\ z^2+1\rangle,\ \langle y+z,\ -z+x,\ z^2+1\rangle]$$